Agilent AN 369-5
Multi-frequency C-V Measurement of Semiconductors

Application Note

Agilent E4980A and 4284A Precision LCR Meters

Introduction

Parameters such as the capacitance of the oxide layer (Cox) and the density of substrate impurities (Nsub) that are required in the evaluation of the manufacturing process of MOS type semiconductors can be derived by using measured C-V characteristics. To make an accurate evaluation of these processes, precise C-V measurements are required. Such measurements entail the following difficulties.

Difficulties

1. There is no single instrument that can make C-V measurements from low to high frequencies.
2. It is difficult to compensate for the additional errors that occur when cable extensions or a prober are used.
3. The accuracy and reliability of the DC bias voltage are not sufficient to perform repeatable C-V measurements.

In the following discussion we will show how C-V measurements are performed with the E4980A and 4284A solving these problems.

Solutions Offered by the E4980A and 4284A

1. Wide Frequency Range Measurements from 20 Hz to 2 MHz.

The program listing shown in the appendix was used to measure the C-V characteristics at 10 kHz, 100 kHz, and 1 MHz of the MOS device whose characteristics are shown in Figure 1.

Thus, the E4980A and the 4284A can single-handedy perform C-V measurements in both the low and high frequency ranges. This allows it to measure high-loss devices (semiconductors on large diameter wafers, etc.), which are difficult to measure at 1 MHz, at low frequencies (10 kHz, 100 kHz, etc.).



Figure 1. Multi-frequency C-V characteristics of a semiconductor

2. Compensation Functions for Cable Extensions and Probers.

In order to test semiconductor devices on a wafer, an extension cable and a prober are required. (See Figure 2 and 3.) The extension cable and the prober cause additional errors that greatly influence the test value. The E4980A and 4284A's 2m/4m Cable Length Operation function with 16048D/E test leads, Option 006 for 4284A, minimizes additional errors caused by using 2m/4m extension cables. The E4980A and 4284A's open/short/load compensation functions minimize errors caused by a prober. This powerful compensation function ensures highly accurate C-V measurements even when a prober is used.



Figure 2. Cable extension and prober connection in four-terminal pair (4TP) configuration



Figure 3. Cable extension and prober connection in two-terminal pair (2TP) configuration

3. Highly Accurate Internal Bias

Insufficient accuracy and stability of the bias voltage applied to a device prevents accurate C-V measurements. The E4980A and 4284A ensure the application of a stable bias voltage with a maximum accuracy of 0.1% (Option 001). This minimizes measurement errors due to bias voltage errors.

Table 1. E4980A Option 001 DC Bias Capability

Range	Resolution	Accuracy*
±0 to 5 V	330 µV	± 0.1% + 2 mV
±5.001 to 10 V	1 mV	± 0.1% + 2 mV
±10.002 to 20 V	2 mV	± 0.1% + 2 mV
±20.005 to 40 V	5 mV	± 0.1% + 2 mV

Table 2. 4284A Option 001 DC Bias Capability

Range	Resolution	Accuracy*
±0.000 to 5 V	1 mV	± 0.1% + 1 mV
±4.002 to 8 V	2 mV	± 0.1% + 2 mV
±8.005 to 20 V	5 mV	± 0.1% + 5 mV
±20.01 to 40 V	10 mV	± 0.1% + 10 mV

*This can only be used when the test signal level is < 2 Vrms

Conclusion

The E4980A and 4284A feature a wide frequency range, powerful compensation functions, and highly accurate bias characteristics. This versatility of the meter guarantees highly accurate C-V measurements, and higher efficiency in the evaluation of semiconductor processes. Because the E4980A and 4284A can perform all of the low and high frequency C-V measurements singlehandedly, it will greatly contribute to the lowering of capital costs.

Appendix 1. Agilent E4980A Sample Program List

```
10 ! ******* Agilent E4980A C-V MEASUREMENT SAMPLE PROGRAM *******
20   OPTION BASE 1
30   GCLEAR
40   PRINT CHR$(12)                                          ! CLEAR DISPLAY
50   DIM C(3,101),Cn(3,101),Vbias(101)
60   INTEGER H1,H2,H3,Tm
70   DIM D(101,4)
80   ASSIGN @Ad TO 717;FORMAT ON                             ! DEFINE I/O PATH
90   ASSIGN @Pat TO 717;FORMAT OFF                           ! DEFINE I/O PATH
100   REMOTE @Ad                                             ! SET E4980A TO REMOTE MODE
110   OUTPUT @Ad;" *Rst"                                     ! RESET E4980A
120   OUTPUT @Ad;"VOLT:LEV 0.02V"                            ! SET OSC LEVEL TO 20mV
130   OUTPUT @Ad;"TRIG:SOUR BUS"                             ! GPIB TRIGGER MODE
140   OUTPUT @Ad;"FORM:DATA REAL"                            ! BINARY DATA TRANSFER
150   OUTPUT @Ad;"MEM:DIM DBUF,101"                          ! DEFINE BUFFER IN E4980A
160   OUTPUT @Ad;"MEM:FILL DBUF"                             ! ENABLE BUFFER TRANSFER
170   OUTPUT @Ad;"APER SHOR"                                 ! SET INTEG TIME TO SHORT
180   GOSUB Corr
190   PRINT "CONNECT OUT AND PRESS CONTINUE"
200   PAUSE
210   PRINT CHR$(12)                                         ! CLEAR DISPLAY
220   Holdtime=10                                            ! HOLD TIME = 10sec
230   Delaytime=.1                                           ! DELAY TIME = 100msec
240   Vbias(1)=-5                                            ! START BIAS VOLTAGE
250   Vstep=.1                                               ! 100mV BIAS STEP
260   OUTPUT @Ad;"TRIG:DEL "&VAL$(Delaytime)                 ! SET DELAY TIME
270   OUTPUT @Ad;"OUTP:DC:ISOL:STAT ON"                      ! DCI ISOLATION ON
280   OUTPUT @Ad;"BIAS:STATE ON"                             ! BIAS ON
290   FreqS(1)=" 1MHz"
300   FreqS(2)=" 100KHz"
310   FreqS(3)=" 10KHz"
320   FOR I=1 TO 3
330     OUTPUT @Ad;"FREQ"&FreqS(I)                           ! FREQUENCY SETTING
340     OUTPUT @Ad;"BIAS:VOLT "&VAL$(Vbias(1))               ! SET START BIAS VOLTAGE
350     PRINT TABXY(5,15);"HOLD TIME 10SEC"
360     WAIT Holdtime                                        ! HOLD TIME
370     PRINT TABXY(5,15);"MEASURING at "&FreqS(I)
380     FOR J=1 TO 101
390       OUTPUT @Ad;"trig"                                  ! TRIGGER E4980A
400       IF J=101 THEN 430
410       Vbias(J+1)=Vbias(J)+Vstep                          ! CHANGE BIAS VOLTAGE
420       OUTPUT @Ad;"BIAS:VOLT "&VAL$(Vbias(J+1)) ! SET BIAS VOLTAGE
430     NEXT J
440     OUTPUT @Ad;"MEM:READ? DBUF"                          ! READ BUFFER
450     ENTER @Pat;H1,H2,H3,D(*),Tm
460      FOR J=1 TO 101
470       C(I,J)=D(J,1)
480     NEXT J
490     IF MAX(C(*))=0 THEN Err                               CHECK IF MAX C VALUE IS 0
500     FOR J=1 TO 101
510       Cn(I,J)=C(I,J)/MAX(C(*))                           ! NORMALIZE C VALUE
520     NEXT J
530     OUTPUT @Ad;"MEM:CLE DBUF;FILL DBUF"                  ! CLEAR DATA IN BUFFER
540     PRINT CHR$(12)                                       ! CLEAR DISPLAY
550   NEXT I
560   OUTPUT @Ad;"BIAS:STATE OFF"                            ! BIAS OFF
570   GOSUB Plotting
580   GOTO Ending
590       !
600   ! ********************** PLOT C-V ***********************
610   ! THE FOLLOWINGS ARE FOR PLOTTING THE MEASUREMENT DATA TO CRT.
620       !
630 Plotting: DEG                                            ! SET ANGLE UNIT TO DEGREE
640   GRAPHICS ON
650   VIEWPORT 8,150,17,100                                  ! PLOT NUMBERS AND LABELS
660   WINDOW 0,100,0,100                                     !
670   LDIR 0                                                 !
680   LORG 2                                                 !
690   CSIZE 4                                                !
700   MOVE 45,15                                             !
710   LABEL "BIAS (V)"                                       !
720   MOVE 8,23                                              !
730   LABEL "-5.0"                                           !
740   MOVE 49,23                                             !
750   LABEL "0"                                              !
760   MOVE 84,23                                             !
770   LABEL "5.0"                                            !
780   MOVE 5,98                                              !
```

Appendix 1. Agilent E4980A Sample Program List *continued...*

```
790   LABEL "1.0"                                                    !
800   MOVE 65,5                                                      !
810   LABEL " 1MHz"                                                  !
820   MOVE 65,10                                                     !
830   LABEL "100kHz"                                                 !
840   MOVE 65,15                                                     !
850   LABEL " 10kHz"                                                 !
860   MOVE 78,5                                                      !
870   DRAW 88,5                                                      !
880   LINE TYPE 8                                                    !
890   MOVE 78,10                                                     !
900   DRAW 88,10                                                     !
910   LINE TYPE 3                                                    !
920   MOVE 78,15                                                     !
930   DRAW 88,15                                                     !
940   LINE TYPE 1                                                    !
950   LDIR 90                                                        !
960   MOVE 3,55                                                      !
970   LABEL "Cap/Cox"                                                !
980   LDIR 0                                                         !
990   MOVE 5,28                                                      !
1000   LABEL "0.0"                                                   !
1010   MOVE 5,98                                                     !
1020   LABEL "1.0"                                                   ! END PLOT NUMBERS AND LABELS
1030                                                                 !
1040   VIEWPORT 25,125,40,95                                         ! DRAW AXES
1050   FRAME                                                         !
1060   WINDOW -5,5,0,1                                               !
1070   AXES 1,.2,-5,0                                                ! END DRAW AXES
1080                                                                 !
1090   FOR I=1 TO 3                                                  ! PLOT DATA
1100      MOVE Vbias(1),Cn(I,1)                                      !
1110      FOR J=2 TO 101                                             !
1120         DRAW Vbias(J),Cn(I,J)                                   !
1130      NEXT J                                                     !
1140      LINE TYPE 3                                                !
1150      IF I=1 THEN LINE TYPE 8                                    !
1160   NEXT I                                                        ! END PLOT DATA
1170   LINE TYPE 1
1180   RETURN
1190      !
1200  Corr:    ! ************* SUBROUTINE FOR CORRECTION *************
1210   ! THE FOLLOWINGS ARE FOR PERFORMING OPEN/SHORT COMPENSATION.
1220      !
1230   OUTPUT @Ad;"DISP:PAGE CSETUP"                                 ! GO TO CORRECTION SETUP PAGE
1240   OUTPUT @Ad;"CORR:LENG 1;METH SING"                            ! CABLE LENGTH 1m, SINGLE COMPEN MODE
1250   OUTPUT @Ad;"CORR:SPOT1:FREQ 1MHz;STAT ON"                     ! SPOT FREQ 1 = 1MHz
1260   OUTPUT @Ad;"CORR:SPOT2:FREQ 100KHz;STAT ON"                   ! SPOT FREQ 2 = 100kHz
1270   OUTPUT @Ad;"CORR:SPOT3:FREQ 10KHz;STAT ON"                    ! SPOT FREQ 3 = 10kHz
1280   OUTPUT @Ad;"CORR:LOAD:STAT OFF"                               ! LOAD COMPEN OFF
1290   PRINT "OPEN COMPENSATION"
1300   PRINT TABXY(5,15);"PRESS CONTINUE"
1310   PAUSE
1320   PRINT TABXY(5,15);"              "
1330   OUTPUT @Ad;"CORR:SPOT1:OPEN"                                  ! PERFORM OPEN COMPEN AT SPOT FREQ 1
1340   OUTPUT @Ad;"CORR:SPOT2:OPEN"                                  ! PERFORM OPEN COMPEN AT SPOT FREQ 2
1350   OUTPUT @Ad;"CORR:SPOT3:OPEN"                                  ! PERFORM OPEN COMPEN AT SPOT FREQ 3
1360   OUTPUT @Ad;"*OPC?"                                            ! CONFIRM OPEN MEASUREMENT COMPLETED
1370   ENTER @Ad;A$
1380   OUTPUT @Ad;"CORR:OPEN:STAT ON"                                ! OPEN COMPEN ON
1390   PRINT CHR$(12)                                                ! CLEAR DISPLAY
1400   PRINT "SHORT COMPENSATION"
1410   PRINT TABXY(5,15);"PRESS CONTINUE"
1420   PAUSE
1430   PRINT TABXY(5,15);"              "
1440   OUTPUT @Ad;"CORR:SPOT1:SHOR"                                  ! PERFORM SHORT COMPEN AT SPOT FREQ 1
1450   OUTPUT @Ad;"CORR:SPOT2:SHOR"                                  ! PERFORM SHORT COMPEN AT SPOT FREQ 2
1460   OUTPUT @Ad;"CORR:SPOT3:SHOR"                                  ! PERFORM SHORT COMPEN AT SPOT FREQ 3
1470   OUTPUT @Ad;"*OPC?"                                            ! CONFIRM SHORT MEASUREMENT COMPLETED
1480   ENTER @Ad;A$
1490   OUTPUT @Ad;"CORR:SHOR:STAT ON"                                ! SHORT COMPEN ON
1500   OUTPUT @Ad;"DISP:PAGE MEAS"                                   ! GO TO MEASUREMENT PAGE
1510   PRINT CHR$(12)                                                ! CLEAR DISPLAY
1520   RETURN
1530      !
1540   ! ***************************************************************
1550  Err: PRINT "C-V MEASUREMENT FAILED."
1560  Ending: END
```

Appendix 2. Agilent 4284A Sample program list

```
10   ! ******* Agilent 4284A C-V MEASUREMENT SAMPLE PROGRAM *******
20   OPTION BASE 1
30   GCLEAR
40   PRINT CHR$(12)                                          ! CLEAR DISPLAY
50   DIM C(3,101),Cn(3,101),Vbias(101)
60   INTEGER H1,H2,H3,Tm
70   DIM D(101,4)
80   ASSIGN @Ad TO 717;FORMAT ON                             ! DEFINE I/O PATH
90   ASSIGN @Pat TO 717;FORMAT OFF                           ! DEFINE I/O PATH
100  REMOTE @Ad                                              ! SET 4284A TO REMOTE MODE
110  OUTPUT @Ad;" *Rst"                                      ! RESET 4284A
120  OUTPUT @Ad;"VOLT:LEV 0.02V"                             ! SET OSC LEVEL TO 20mV
130  OUTPUT @Ad;"OUTP:HPOW ON"                               ! BIAS OPTION ON
140  OUTPUT @Ad;"TRIG:SOUR BUS"                              ! GPIB TRIGGER MODE
150  OUTPUT @Ad;"FORM:DATA REAL"                             ! BINARY DATA TRANSFER
160  OUTPUT @Ad;"MEM:DIM DBUF,101"                           ! DEFINE BUFFER IN 4284A
170  OUTPUT @Ad;"MEM:FILL DBUF"                              ! ENABLE BUFFER TRANSFER
180  OUTPUT @Ad;"APER SHOR"                                  ! SET INTEG TIME TO SHORT
190  GOSUB Corr
200  PRINT "CONNECT OUT AND PRESS CONTINUE"
210  PAUSE
220  PRINT CHR$(12)                                          ! CLEAR DISPLAY
230  Holdtime=10                                             ! HOLD TIME = 10sec
240  Delaytime=.1                                            ! DELAY TIME = 100msec
250  Vbias(1)=-5                                             ! START BIAS VOLTAGE
260  Vstep=.1                                                ! 100mV BIAS STEP
270  OUTPUT @Ad;"TRIG:DEL "&VAL$(Delaytime)                  ! SET DELAY TIME
280  OUTPUT @Ad;"OUTP:DC:ISOL ON"                            ! DC BIAS ISOLATION ON
290  OUTPUT @Ad;"BIAS:STATE ON"                              ! BIAS ON
300  FreqS(1)=" 1MHz"
310  FreqS(2)=" 100KHz"
320  FreqS(3)=" 10KHz"
330  FOR I=1 TO 3
340     OUTPUT @Ad;"FREQ"&FreqS(I)                           ! FREQUENCY SETTING
350     OUTPUT @Ad;"BIAS:VOLT "&VAL$(Vbias(1))               ! SET START BIAS VOLTAGE
360     PRINT TABXY(5,15);"HOLD TIME 10SEC"
370     WAIT Holdtime                                        ! HOLD TIME
380     PRINT TABXY(5,15);"MEASURING at "&FreqS(I)
390     FOR J=1 TO 101
400        OUTPUT @Ad;"trig"                                 ! TRIGGER 4284A
410        IF J=101 THEN 440
420        Vbias(J+1)=Vbias(J)+Vstep                         ! CHANGE BIAS VOLTAGE
430        OUTPUT @Ad;"BIAS:VOLT "&VAL$(Vbias(J+1))          ! SET BIAS VOLTAGE
440     NEXT J
450     OUTPUT @Ad;"MEM:READ? DBUF"                          ! READ BUFFER
460     ENTER @Pat;H1,H2,H3,D(*),Tm
470      FOR J=1 TO 101
480        C(I,J)=D(J,1)
490     NEXT J
500     IF MAX(C(*))=0 THEN Err                              ! CHECK IF MAX C VALUE IS 0
510     FOR J=1 TO 101
520        Cn(I,J)=C(I,J)/MAX(C(*))                          ! NORMALIZE C VALUE
530         !PRINT Cn(I,J)                                   ! PRINT NORMALIZED C VALUE
540     NEXT J
550     OUTPUT @Ad;"MEM:CLE DBUF;FILL DBUF"                  ! CLEAR DATA IN BUFFER
560     PRINT CHR$(12)                                       ! CLEAR DISPLAY
570  NEXT I
580  OUTPUT @Ad;"BIAS:STATE OFF"                             ! BIAS OFF
590  GOSUB Plotting
600  GOTO Ending
610  !
620  ! ********************** PLOT C-V **********************
630  ! THE FOLLOWINGS ARE FOR PLOTTING THE MEASUREMENT DATA TO CRT.
640  !
650  Plotting: DEG                                           ! SET ANGLE UNIT TO DEGREE
660   GRAPHICS ON
670   VIEWPORT 8,150,17,100                                  ! PLOT NUMBERS AND LABELS
680   WINDOW 0,100,0,100                                     !
690   LDIR 0                                                 !
700   LORG 2                                                 !
710   CSIZE 4                                                !
720   MOVE 45,15                                             !
730   LABEL "BIAS (V)"                                       !
740   MOVE 8,23                                              !
750   LABEL "-5.0"                                           !
760   MOVE 49,23                                             !
770   LABEL "0"                                              !
780   MOVE 84,23                                             !
790   LABEL "5.0"                                            !
```

Appendix 2. Agilent 4284A Sample program list *continued...*

```
800   MOVE 5,98                                                    !
810   LABEL "1.0"                                                  !
820   MOVE 65,5                                                    !
830   LABEL " 1MHz"                                                !
840   MOVE 65,10                                                   !
850   LABEL "100kHz"                                               !
860   MOVE 65,15                                                   !
870   LABEL " 10kHz"                                               !
880   MOVE 78,5                                                    !
890   DRAW 88,5                                                    !
900   LINE TYPE 8                                                  !
910   MOVE 78,10                                                   !
920   DRAW 88,10                                                   !
930   LINE TYPE 3                                                  !
940   MOVE 78,15                                                   !
950   DRAW 88,15                                                   !
960   LINE TYPE 1                                                  !
970   LDIR 90                                                      !
980   MOVE 3,55                                                    !
990   LABEL "Cap/Cox"                                              !
1000   LDIR 0                                                      !
1010   MOVE 5,28                                                   !
1020   LABEL "0.0"                                                 !
1030   MOVE 5,98                                                   !
1040   LABEL "1.0"                                                 ! END PLOT NUMBERS AND LABELS
1050   !
1060   VIEWPORT 25,125,40,95                                       ! DRAW AXES
1070   FRAME                                                       !
1080   WINDOW -5,5,0,1                                             !
1090   AXES 1,.2,-5,0                                              ! END DRAW AXES
1100   !
1110   FOR I=1 TO 3                                                ! PLOT DATA
1120      MOVE Vbias(1),Cn(I,1)                                    !
1130      FOR J=2 TO 101                                           !
1140         DRAW Vbias(J),Cn(I,J)                                 !
1150      NEXT J                                                   !
1160      LINE TYPE 3                                              !
1170      IF I=1 THEN LINE TYPE 8                                  !
1180   NEXT I                                                      ! END PLOT DATA
1190   LINE TYPE 1
1200   RETURN
1210      !
1220   Corr:    ! ************* SUBROUTINE FOR CORRECTION *************
1230      ! THE FOLLOWINGS ARE FOR PERFORMING OPEN/SHORT COMPENSATION.
1240      !
1250   OUTPUT @Ad;"DISP:PAGE CSETUP"                               ! GO TO CORRECTION SETUP PAGE
1260   OUTPUT @Ad;"CORR:LENG 1;METH SING"                          ! CABLE LENGTH 1m, SINGLE COMPEN MODE
1270   OUTPUT @Ad;"CORR:SPOT1:FREQ 1MHz;STAT ON"                   ! SPOT FREQ 1 = 1MHz
1280   OUTPUT @Ad;"CORR:SPOT2:FREQ 100KHz;STAT ON"                 ! SPOT FREQ 2 = 100kHz
1290   OUTPUT @Ad;"CORR:SPOT3:FREQ 10KHz;STAT ON"                  ! SPOT FREQ 3 = 10kHz
1300   OUTPUT @Ad;"CORR:LOAD:STAT OFF"                             ! LOAD COMPEN OFF
1310   PRINT "OPEN COMPENSATION"
1320   PRINT TABXY(5,15);"PRESS CONTINUE"
1330   PAUSE
1340   PRINT TABXY(5,15);"              "
1350   OUTPUT @Ad;"CORR:SPOT1:OPEN"                                ! PERFORM OPEN COMPEN AT SPOT FREQ 1
1360   OUTPUT @Ad;"CORR:SPOT2:OPEN"                                ! PERFORM OPEN COMPEN AT SPOT FREQ 2
1370   OUTPUT @Ad;"CORR:SPOT3:OPEN"                                ! PERFORM OPEN COMPEN AT SPOT FREQ 3
1380   OUTPUT @Ad;"*OPC?"                                          ! CONFIRM OPEN MEASUREMENT COMPLETED
1390   ENTER @Ad;A$
1400   OUTPUT @Ad;"CORR:OPEN:STAT ON"                              ! OPEN COMPEN ON
1410   PRINT CHR$(12)                                              ! CLEAR DISPLAY
1420   PRINT "SHORT COMPENSATION"
1430   PRINT TABXY(5,15);"PRESS CONTINUE"
1440   PAUSE
1450   PRINT TABXY(5,15);"              "
1460   OUTPUT @Ad;"CORR:SPOT1:SHOR"                                ! PERFORM SHORT COMPEN AT SPOT FREQ 1
1470   OUTPUT @Ad;"CORR:SPOT2:SHOR"                                ! PERFORM SHORT COMPEN AT SPOT FREQ 2
1480   OUTPUT @Ad;"CORR:SPOT3:SHOR"                                ! PERFORM SHORT COMPEN AT SPOT FREQ 3
1490   OUTPUT @Ad;"*OPC?"                                          ! CONFIRM SHORT MEASUREMENT COMPLETED
1500   ENTER @Ad;A$
1510   OUTPUT @Ad;"CORR:SHOR:STAT ON"                              ! SHORT COMPEN ON
1520   OUTPUT @Ad;"DISP:PAGE MEAS"                                 ! GO TO MEASUREMENT PAGE
1530   PRINT CHR$(12)                                              ! CLEAR DISPLAY
1540   RETURN
1550   !
1560   ! ***************************************************************
1570 Err: PRINT "C-V MEASUREMENT FAILED."
1580 Ending: END
```

www.agilent.com

Agilent Email Updates

www.agilent.com/find/emailupdates
Get the latest information on the products and applications you select.

Agilent Direct

www.agilent.com/find/agilentdirect
Quickly choose and use your test equipment solutions with confidence.

Agilent Open

www.agilent.com/find/open
Agilent Open simplifies the process of connecting and programming test systems to help engineers design, validate and manufacture electronic products. Agilent offers open connectivity for a broad range of system-ready instruments, open industry software, PC-standard I/O and global support, which are combined to more easily integrate test system development.

Remove all doubt

Our repair and calibration services will get your equipment back to you, performing like new, when promised. You will get full value out of your Agilent equipment throughout its lifetime. Your equipment will be serviced by Agilent-trained technicians using the latest factory calibration procedures, automated repair diagnostics and genuine parts. You will always have the utmost confidence in your measurements. For information regarding self maintenance of this product, please contact your Agilent office.

Agilent offers a wide range of additional expert test and measurement services for your equipment, including initial start-up assistance, onsite education and training, as well as design, system integration, and project management.

For more information on repair and calibration services, go to:

www.agilent.com/find/removealldoubt

Product specifications and descriptions in this document subject to change without notice.

For more information on Agilent Technologies' products, applications or services, please contact your local Agilent office. The complete list is available at:

www.agilent.com/find/contactus

Americas

Canada	(877) 894-4414
Latin America	305 269 7500
United States	(800) 829-4444

Asia Pacific

Australia	1 800 629 485
China	800 810 0189
Hong Kong	800 938 693
India	1 800 112 929
Japan	0120 (421) 345
Korea	080 769 0800
Malaysia	1 800 888 848
Singapore	1 800 375 8100
Taiwan	0800 047 866
Thailand	1 800 226 008

Europe & Middle East

Austria	01 36027 71571
Belgium	32 (0) 2 404 93 40
Denmark	45 70 13 15 15
Finland	358 (0) 10 855 2100
France	0825 010 700*
	*0.125 €/minute
Germany	07031 464 6333
Ireland	1890 924 204
Israel	972-3-9288-504/544
Italy	39 02 92 60 8484
Netherlands	31 (0) 20 547 2111
Spain	34 (91) 631 3300
Sweden	0200-88 22 55
Switzerland	0800 80 53 53
United Kingdom	44 (0) 118 9276201

Other European Countries:
www.agilent.com/find/contactus

Revised: October 6, 2008

Printed in USA, December 10, 2008
5950-2953

Agilent Technologies